

Chapel: Data Parallelism

Data vs. Task Parallelism (Our Definitions)

Data Parallelism:

- parallelism is driven by collections of data
 - data aggregates (arrays)
 - sets of indices (ranges, domains)
 - other user-defined collections
- e.g., “for all elements in array A ...”

Task Parallelism:

- parallelism is expressed in terms of distinct computations
- e.g., “create a task to do foo() while another does bar()”

(Of course, data parallelism is executed using tasks and task parallelism typically operates on data, so the line can get fuzzy at times...)



"Hello World" in Chapel: a Data Parallel Version

- Data Parallel Hello World

```
config const numIters = 100000;

forall i in 1..numIters do
  writeln("Hello, world! ",
    "from iteration ", i, " of ", numIters);
```

Outline

- Domains and Arrays
 - Rectangular Domains and Arrays
 - Iterations and Operations
- Other Domain Types
- Reductions and Scans
- Jacobi Iteration Example



Domains

Domain: A first-class index set


- A fundamental Chapel concept for data parallelism
- Domains may optionally be distributed



Sample Domains

```
config const m = 4, n = 8;

var D: domain(2) = {1..m, 1..n};
```




D

Sample Domains

```
config const m = 4, n = 8;

var D: domain(2) = {1..m, 1..n};

var Inner: subdomain(D) = {2..m-1, 2..n-1};
```



Domains Define Arrays

- Syntax


```
array-type:
  [ domain-expr ] elt-type
```

- Semantics


- Stores an *elt-type* for each index in *domain-expr*

- Example

```
var A, B: [D] real;
```



A



B

- Earlier example, revisited

```
var A: [1..3, 1..5] real; // [1..3, 1..5] creates an
                        // anonymous domain
```


Domain Iteration

- For loops (discussed already)
 - Execute loop body once per domain index, serially

```
for i in Inner do ...
```

	1	2	3	4	5	6	
	7	8	9	10	11	12	

- Forall loops

- Executes loop body once per domain index, in parallel
- Loop must be *serializable* (executable by one task)

```
forall i in Inner do ...
```

- Loop variables take on `const` domain index values

Other Forall Loops

Forall loops also support...

- A shorthand notation:

```
[ (i,j) in D ] A[i,j] = i + j/10.0;
```

- Expression-based forms:

```
A = forall (i,j) in D do i + j/10.0;
```

```
A = [ (i,j) in D ] i + j/10.0;
```

1.1	1.2	1.3	1.4	1.5	1.6	1.7	1.8
2.1	2.2	2.3	2.4	2.5	2.6	2.7	2.8
3.1	3.2	3.3	3.4	3.5	3.6	3.7	3.8
4.1	4.2	4.3	4.4	4.5	4.6	4.7	4.8



A

Domain Algebra


Domain values support...

- Methods for creating new domains

```
var D2 = Inner.expand(1, 0);
```




```
var D3 = Inner.translate(0, 1);
```



- Intersection via Slicing

```
var D4 = D2[D3];
```




- Range operators (e.g., #, by, align)

Sub-Arrays/Array Slicing

Indexing into arrays with domain values results in a sub-array expression (an “array slice”)

```
A[Inner] = B[Inner.translate(0,1)];
```




Array Reallocation


Reassigning a domain logically reallocates its arrays

- array values are preserved for common indices

$D = \{1..2*m, 1..2*n\};$



A



B

Array Iteration

- Array expressions also support for and forall loops

```
for a in A[Inner] do ...
```

	1	2	3	4	5	6	
	7	8	9	10	11	12	

```
forall a in A[Inner] do ...
```

- Array loop indices refer to array elements (can be modified)

```
forall (a, (i,j)) in zip(A, D) do a = i + j/10.0;
```

Note that forall loops support zippered iteration, like for-loops

1.1	1.2	1.3	1.4	1.5	1.6	1.7	1.8
2.1	2.2	2.3	2.4	2.5	2.6	2.7	2.8
3.1	3.2	3.3	3.4	3.5	3.6	3.7	3.8
4.1	4.2	4.3	4.4	4.5	4.6	4.7	4.8

Array Indexing

- Arrays can be indexed using variables of their domain's index type (tuples) or lists of integers

```
var i = 1, j = 2;  
var ij = (i, j);  
  
A[ij] = 1.0;  
A[i, j] = 2.0;
```

- Array indexing can use either parentheses or brackets

```
A(ij) = 3.0;  
A(i, j) = 4.0;
```

Array Arguments and Aliases

- Arrays are passed by reference by default

```
proc zero(X: []) { X = 0; }

zero(A[Inner]);  // zeroes the inner values of A
```

- Formal array arguments can reindex actuals

```
proc f(X: [1..b,1..b]) { ... } // X uses 1-based indices

f(A[lo..#b, lo..#b]);
```

- Array alias declarations provide similar functionality

```
var InnerA => A[Inner];
var InnerA1: [1..n-2,1..m-2] => A[2..n-1,2..m-1];
```


Promoting Functions and Operators

Functions/operators expecting scalars can also take...
 ...arrays, causing each element to be passed in

<code>sin(A)</code> <code>2*A</code>	≈	<code>forall a in A do sin(a)</code> <code>forall a in A do 2*a</code>
---	---	---

...domains, causing each index to be passed in

<code>foo(Inner)</code>	≈	<code>forall i in Inner do foo(i)</code>
-------------------------	---	--

Multiple arguments promote using zippered iteration

<code>pow(A, B)</code>	≈	<code>forall (a,b) in zip(A,B) do pow(a,b)</code>
------------------------	---	---

Data Parallelism is Implicit

- forall loops are implemented using multiple tasks
 - ditto for operations that are equivalent to foralls
 - details depend on what is being iterated over
- many times, this parallelism can seem invisible
 - for this reason, Chapel's data parallelism can be considered *implicitly parallel*
 - it also tends to make the data parallel features easier to use and less likely to result in bugs as compared to explicit tasks



How Much Parallelism?

By default*, controlled by three config variables:

--dataParTasksPerLocale=#

- Specify # of tasks to execute forall loops
- *Current Default:* number of processor cores

--dataParIgnoreRunningTasks=[true | false]

- If false, reduce # of forall tasks by # of running tasks
- *Current Default:* true

--dataParMinGranularity=#

- If > 0, reduce # of forall tasks if any task has fewer iterations
- *Current Default:* 1

*Default values can be overridden for specific domains/arrays



Outline


- Domains and Arrays
- Other Domain Types
 - Strided
 - Sparse
 - Associative
 - Opaque
- Reductions and Scans
- Jacobi Iteration Example




Chapel Domain Types

Chapel supports several domain types...


```
var OceanSpace = {0..#lat, 0..#long},
    AirSpace = OceanSpace by (2,4),
    IceSpace: sparse subdomain(OceanSpace) = genCaps();
```



dense




strided




sparse

unstructured



associative








```
var Vertices: domain(opaque) = ...,   People: domain(string) = ...;
```

Chapel Array Types

All domain types can be used to declare arrays...

```
var Ocean: [OceanSpace] real,
    Air: [AirSpace] real,
    IceCaps[IceSpace] real;
```








```
var Weight: [Vertices] real,
```

```
Age: [People] int;
```

Iteration

...to iterate over index sets...

```
forall ij in AirSpace do
    Ocean[ij] += IceCaps[ij];
```








```
forall v in Vertices do
    Weight[v] = numEdges[v];
```

```
forall p in People do
    Age[p] += 1;
```

Slicing

...to slice arrays...

```
Ocean[AirSpace] += IceCaps[AirSpace];
```



...Vertices[Interior]...





...People[Interns]...

Reallocation

...and to reallocate arrays

```
AirSpace = OceanSpace by (2,2);
```

```
IceSpace += genEquator();
```



	"steve"
	"lee"
	"sung"
	"david"
	"jacob"
	"albert"
	"brad"
	"vass"

```
newnode = Vertices.create();
```

```
People += "vass";
```

Associative Domains and Arrays by Example

```

var Presidents: domain(string) =
    {"George", "John", "Thomas",
     "James", "Andrew", "Martin"};

Presidents += "William";

var Age: [Presidents] int,
    Birthday: [Presidents] string;

Birthday["George"] = "Feb 22";

forall president in President do
    if Birthday[president] == today then
        Age[president] += 1;
  
```

George
John
Thomas
James
Andrew
Martin
William

Presidents

Feb 22
Oct 30
Apr 13
Mar 16
Mar 15
Dec 5
Feb 9

Birthday

277
274
266
251
242
227
236

Age

Outline

- Domains and Arrays
- Other Domain Types
- Reductions and Scans
- Jacobi Iteration Example



Reductions

- Syntax

```
reduce-expr:
  reduce-op reduce iterator-expr
```

- Semantics

- Combines argument values using *reduce-op*
- *Reduce-op* may be built-in or user-defined

- Examples

```
total = + reduce A;
bigDiff = max reduce [i in Inner] abs (A[i]-B[i]);
(minVal, minLoc) = minloc reduce zip (A, D);
```



Scans

- Syntax

```
scan-expr:
  scan-op scan iterator-expr
```

- Semantics

- Computes parallel prefix over values using *scan-op*
- Scan-op* may be any *reduce-op*

- Examples

```
var A, B, C: [1..5] int;  
A = 1; // A: 1 1 1 1 1  
B = + scan A; // B: 1 2 3 4 5  
B[3] = -B[3]; // B: 1 2 -3 4 5  
C = min scan B; // C: 1 1 -3 -3 -3
```

Reduction and Scan Operators


- Built-in
 - $+$, $*$, $\&\&$, $||$, $\&$, $|$, $^$, min, max
 - minloc, maxloc
 - Takes a zipped pair of values and indices
 - Generates a tuple of the min/max value and its index
- User-defined
 - Defined via a class that implements a standard interface
 - Compiler generates code that calls these methods




Outline

- Domains and Arrays
- Other Domain Types
- Reductions and Scans
- Jacobi Iteration Example

Jacobi Iteration in Pictures



repeat until max
change $< \epsilon$



Jacobi Iteration in Chapel

```
config const n = 6,  
            epsilon = 1.0e-5;  
  
const BigD: domain(2) = {0..n+1, 0..n+1},  
      D: subdomain(BigD) = {1..n, 1..n},  
      LastRow: subdomain(BigD) = D.exterior(1,0);  
  
var A, Temp : [BigD] real;  
  
A[LastRow] = 1.0;  
  
do {  
  [(i,j) in D] Temp[i,j] = (A[i-1,j] + A[i+1,j]  
                             + A[i,j-1] + A[i,j+1]) / 4;  
  
  const delta = max reduce abs(A[D] - Temp[D]);  
  A[D] = Temp[D];  
} while (delta > epsilon);  
  
writeln(A);
```



Jacobi Iteration in Chapel

```

config const n = 6,
              epsilon = 1.0e-5;

const BigD: domain(2) = {0..n+1, 0..n+1},
      D: subdomain(BigD) = {1..n, 1..n},
      LastRow: subdomain(BigD) = D.exterior(1,0);

var A, Temp : [BigD] real;

```

Declare program parameters

const \Rightarrow can't change values after initialization

config \Rightarrow can be set on executable command-line

prompt> jacobi --n=10000 --epsilon=0.0001

note that no types are given; inferred from initializer

n \Rightarrow **default integer** (32 bits)

epsilon \Rightarrow **default real floating-point** (64 bits)

A[Las

do {
[(i

con
A[D
} **whi**

writeln(A);



Jacobi Iteration in Chapel

```

config const n = 6,
            epsilon = 1.0e-5;


const BigD: domain(2) = {0..n+1, 0..n+1},
      D: subdomain(BigD) = {1..n, 1..n},
      LastRow: subdomain(BigD) = D.exterior(1,0);

```

Declare domains (first class index sets)

domain(2) \Rightarrow 2D arithmetic domain, indices are integer 2-tuples

subdomain(*P*) \Rightarrow a domain of the same type as *P* whose indices are guaranteed to be a subset of *P*'s



exterior \Rightarrow one of several built-in domain generators



Jacobi Iteration in Chapel

```
config const n = 6,
            epsilon = 1.0e-5;

const BigD: domain(2) = {0..n+1, 0..n+1},
      D: subdomain(BigD) = {1..n, 1..n},
      LastRow: subdomain(BigD) = D.exterior(1,0);
```

```
var A, Temp : [BigD] real;
```


```
A[LastRow] = 1.0;
```

Declare arrays


var \Rightarrow can be modified throughout its lifetime

: **[BigD]** *T* \Rightarrow array of size *BigD* with elements of type *T*


(**no initializer**) \Rightarrow values initialized to default value (0.0 for reals)



BigD



A



Temp

4;



Jacobi Iteration in Chapel

```

config const n = 6,
            epsilon = 1.0e-5;

const BigD: domain(2) = {0..n+1, 0..n+1},
      D: subdomain(BigD) = {1..n, 1..n},
      LastRow: subdomain(BigD) = D.exterior(1,0);


var A, Temp : [BigD] real;

A[LastRow] = 1.0;

```

Set Explicit Boundary Condition

indexing by domain \Rightarrow slicing mechanism
 array expressions \Rightarrow parallel evaluation




A

Jacobi Iteration in Chapel

```
config const n = 6,  
            epsilon = 1.0e-5;
```

Compute 5-point stencil

$[(i,j) \text{ in } D] \Rightarrow$ parallel forall expression over D 's indices, binding them to new variables i and j



```
[(i,j) in D] Temp[i,j] = (A[i-1,j] + A[i+1,j]  
                        + A[i,j-1] + A[i,j+1]) / 4;
```

```
const delta = max reduce abs(A[D] - Temp[D]);  
A[D] = Temp[D];  
} while (delta > epsilon);  
  
writeln(A);
```

Jacobi Iteration in Chapel

```
config const n = 6,  
            epsilon = 1.0e-5;
```

```
const BigD: domain(2) = {0..n+1, 0..n+1},
```

Compute maximum change

op reduce \Rightarrow collapse aggregate expression to scalar using *op*

Promotion: *abs()* and $-$ are scalar operators, automatically promoted to work with array operands

```
do {  
  [(i,j) in D] Temp[i,j] = (A[i-1,j] + A[i+1,j]  
                           + A[i,j-1] + A[i,j+1]) / 4;  
  
  const delta = max reduce abs(A[D] - Temp[D]);  
  A[D] = Temp[D];  
} while (delta > epsilon);  
  
writeln(A);
```



Jacobi Iteration in Chapel

```

config const n = 6,
               epsilon = 1.0e-5;

const BigD: domain(2) = {0..n+1, 0..n+1},
        D: subdomain(BigD) = {1..n, 1..n},
        LastRow: subdomain(BigD) = D.exterior(1,0);

var A[LastRow, D];

    do {
        [(i,j) in D] Temp[i,j] = (A[i-1,j] + A[i+1,j]
                                   + A[i,j-1] + A[i,j+1]) / 4;

        const delta = max reduce abs(A[D] - Temp[D]);
        A[D] = Temp[D];
    } while (delta > epsilon);

writeln(A);

```

Copy data back & Repeat until done

A[LastRow] uses slicing and whole array assignment
standard *do...while* loop construct



Jacobi Iteration in Chapel

```
config const n = 6,  
            epsilon = 1.0e-5;  
  
const BigD: domain(2) = {0..n+1, 0..n+1},  
      D: subdomain(BigD) = {1..n, 1..n},  
      LastRow: subdomain(BigD) = D.exterior(1,0);  
  
var A, Temp : [BigD] real;  
  
A[LastRow] = 1.0;  
  
do {  
  [(i,j) in D] Temp[i,j] = (A[i-1,j] + A[i+1,j]  
                           + A[i,j-1] + A[i,j+1]) / 4;  
  
  const delta = max reduce abs(A[D] - Temp[D]);  
  A[D] = Temp[D];  
} while (delta > epsilon);  
  
writeln(A);
```

Write array to console



Jacobi Iteration in Chapel

```

config const n = 6,
                epsilon = 1.0e-5;

const BigD = {0..n+1, 0..n+1} dmapped Block(...),
          D: subdomain(BigD) = {1..n, 1..n},
          LastRow: subdomain(BigD) = D.exterior(1,0);

var A, Temp : [BigD] real;


```

With this change, same code runs in a distributed manner

Domain distribution maps indices to *locales*

⇒ decomposition of arrays & default mapping of iterations to locales

Subdomains inherit parent domain's distribution



Jacobi Iteration in Chapel

```
config const n = 6,  
            epsilon = 1.0e-5;  
  
const BigD = {0..n+1, 0..n+1} dmapped Block(...),  
            D: subdomain(BigD) = {1..n, 1..n},  
            LastRow: subdomain(BigD) = D.exterior(1,0);  
  
var A, Temp : [BigD] real;  
  
A[LastRow] = 1.0;  
  
do {  
    [(i,j) in D] Temp[i,j] = (A[i-1,j] + A[i+1,j]  
                             + A[i,j-1] + A[i,j+1]) / 4;  
  
    const delta = max reduce abs(A[D] - Temp[D]);  
    A[D] = Temp[D];  
} while (delta > epsilon);  
  
writeln(A);
```



Data Parallelism: Status

- Most features implemented and working correctly
- Scalar performance not optimal for higher-dimensional domain/array operations
- Implementation of unstructured domains/arrays is correct but inefficient

Future Directions

- Gain more experience with unstructured (graph-based) domains and arrays

Questions?

- Domains and Arrays
 - Regular Domains and Arrays
 - Iterations and Operations
- Other Domain Types
 - Strided
 - Sparse
 - Associative
 - Opaque
- Data Parallel Operations
 - Reductions
 - Scans
- Jacobi Iteration Example